
Monads in programming language theory

Monads are a technical device with several uses in programming:

To structure denotational semantics and make them easy to extend
with new language features. (E. Moggi, 1989.)
Not treated in this lecture.

To factor out commonalities between many program transformations
and between their proofs of correctness.

As a powerful programming techniques in pure functional languages.
(P. Wadler, 1992; the Haskell community).

Introduction to monads

Commonalities between program transformations

Consider the conversions to exception-returning style, state-passing style,
and continuation-passing style. For constants, variables and
λ-abstractions, we have:

[[N]] = V (N) [[N]] = λs.(N, s) [[N]] = λk .k N

[[x ]] = V (x) [[x ]] = λs.(x , s) [[x ]] = λk .k x

[[λx .a]] = V (λx .[[a]]) [[λx .a]] = λs.(λx .[[a]], s) [[λx .a]] = λk .k (λx .[[a]])

in all three cases, we return (put in some appropriate wrapper) the values
N or x or λx .[[a]].

Introduction to monads

Commonalities between program transformations

For let bindings, we have:

[[let x = a in b]] = match [[a]] with E (x)→ E (x) | V (x)→ [[b]]

[[let x = a in b]] = λs. match [[a]] s with (x , s ′)→ [[b]] s ′

[[let x = a in b]] = λk . [[a]] (λx . [[b]] k)

In all three cases, we extract (one way or another) the value contained in
the computation [[a]], bind it to the variable x , and proceed with the
computation [[b]].

Introduction to monads

Commonalities between program transformations

Concerning function applications:

[[a b]] = match [[a]] with

| E (ea)→ E (ea)

| V (va)→

match [[b]] with E (eb)→ E (eb) | V (vb)→ va vb

[[a b]] = λs. match [[a]] s with (va, s
′)→

match [[b]] s ′ with (vb, s
′′)→ va vb s ′′

[[a b]] = λk . [[a]] (λva. [[b]] (λvb. va vb k))

We bind [[a]] to a variable va, then bind [[b]] to a variable vb, then perform
the application va vb.

Introduction to monads

Interface of a monad
A monad is defined by a parameterized type α mon and operations ret,
bind and run, with types:

ret : ∀α. α→ α mon

bind : ∀α, β. α mon→ (α→ β mon)→ β mon

run : ∀α. α mon→ α

The type τ mon is the type of computations that eventually produce a
value of type τ .

ret a encapsulates a pure expression a : τ as a trivial computation (of
type τ mon) that immediately produces the value of a.

bind a (λx .b) performs the computation a : τ mon, binds its value to
x : τ , then performs the computation b : τ ′ mon.

run a is the execution of a whole monadic program a, extracting its return
value.

X. Leroy (INRIA) Functional programming languages MPRI 2-4-2, 2007 7 / 52

Introduction to monads

Monadic laws

The ret and bind operations of the monad are supposed to satisfy the
following algebraic laws:

bind (ret a) f ≈ f a

bind a (λx . ret x) ≈ a

bind (bind a (λx .b)) (λy .c) ≈ bind a (λx . bind b (λy .c))

The relation ≈ needs to be made more precise, but intuitively means
“behaves identically”.

Introduction to monads

Example: the Exception monad
(also called the Error monad)

type α mon = V of α | E of exn

ret a = V(a)

bind m f = match m with E(x) -> E(x) | V(x) -> f x

run m = match m with V(x) -> x

bind encapsulates the propagation of exceptions in compound expressions
such as a b or let bindings.

Additional operations in this monad:

raise x = E(x)

trywith m f = match m with E(x) -> f x | V(x) -> V(x)

Introduction to monads

Example: the State monad

type α mon = state → α× state

ret a = λs. (a, s)

bind m f = λs. match m s with (x, s’) -> f x s’

run m = match m empty_store with (x, s) -> x

bind encapsulates the threading of the state in compound expressions.

Additional operations in this monad:

ref x = λs. store_alloc x s

deref r = λs. (store_read r s, s)

assign r x = λs. store_write r x s

Introduction to monads

Example: the Continuation monad

type α mon = (α→ answer)→ answer

ret a = λk. k a

bind m f = λk. m (λv. f v k)

run m = m (λx. x)

Additional operations in this monad:

callcc f = λk. f k k

throw x y = λk. x y

The monadic translation

Outline

1 Introduction to monads

2 The monadic translation
Definition
Correctness
Application to some monads

3 Monadic programming
More examples of monads
Monad transformers

X. Leroy (INRIA) Functional programming languages MPRI 2-4-2, 2007 12 / 52



The monadic translation Definition

The monadic translation
Core constructs

[[N]] = ret N

[[x ]] = ret x

[[λx .a]] = ret (λx .[[a]])

[[let x = a in b]] = bind [[a]] (λx .[[b]])

[[a b]] = bind [[a]] (λva. bind [[b]] (λvb. va vb))

These translation rules are shared between all monads.

Effect on types: if a : τ then [[a]] : [[τ ]] mon
where [[τ1 → τ2]] = τ1 → [[τ2]] mon and [[τ ]] = τ for base types τ .

The monadic translation Definition

The monadic translation
Extensions

[[µf .λx .a]] = ret (µf .λx .[[a]])

[[a op b]] = bind [[a]] (λva. bind [[b]] (λvb. ret (va op vb)))

[[C (a1, . . . , an)]] = bind [[a1]] (λv1. . . .

bind [[an]] (λvn. ret(C (v1, . . . , vn))))

[[match a with . . . pi . . .]] = bind [[a]] (λva. match va with . . . [[pi ]] . . .)

[[C (x1, . . . , xn)→ a]] = C (x1, . . . , xn)→ [[a]]

The monadic translation Definition

Example of monadic translation

[[1 + f x ]] =

bind (ret 1) (λv1.

bind (bind (ret f) (λv2.

bind (ret x) (λv3. v2 v3))) (λv4.

ret (v1 + v4)))

After administrative reductions using the first monadic law:

[[1 + f x ]] =

bind (f x) (λv. ret (1 + v))

The monadic translation Definition

Example of monadic translation

[[µfact. λn. if n = 0 then 1 else n * fact(n-1)]] =

ret (µfact. λn.

if n = 0

then ret 1

else bind (fact(n-1)) (λv. ret (n * v))

The monadic translation Definition

The monadic translation
Monad-specific constructs and operations

Most additional constructs for exceptions, state and continuations can be
treated as regular function applications of the corresponding additional
operations of the monad. For instance, in the case of raise a:

[[raise a]] = bind (ret raise) (λvr .bind [[a]] (λva. vr va))
adm
→ bind [[a]] (λva. raise va)

The bind takes care of propagating exceptions raised in a.

The only case where we need a special translation rule is the the
try. . . with construct:

[[try a with x → b]] = trywith [[a]] (λx .[[b]])

The monadic translation Correctness

Syntactic properties of the monadic translation

Define the monadic translation of a value [[v ]]v as follows:

[[N]]v = N [[λx .a]]v = λx .[[a]]

Lemma 1 (Translation of values)

[[v ]] = ret [[v ]]v for all values v . Moreover, [[v ]]v is a value.

Lemma 2 (Monadic substitution)

[[a[x ← v ]]] = [[a]][x ← [[v ]]v ] for all values v ,

The monadic translation Correctness

Reasoning about reductions of the translations

If a reduces, is it the case that the translation [[a]] reduces? This depends
on the monad:

For the exception monad, this is true.

For the state and continuation monads, [[a]] is a λ-abstraction which
cannot reduce.

To reason about the evaluation of [[a]], we need in general to put this term
in an appropriate context, for instance

For the state monad: [[a]] s where s is a store value.

For the continuation monad: [[a]] k where k is a continuation λx . . .

The monadic translation Correctness

Contextual equivalence

To overcome this problem, we assume that the monad defines an
equivalence relation a ≈ a′ between terms, which is reflexive, symmetric
and transitive, and satisfies the following properties:

1 (λx .a) v ≈ a[x ← v ]

2 bind (ret v) (λx .b) ≈ b[x ← v ]

3 bind a (λx .b) ≈ bind a′ (λx .b) if a ≈ a′

4 If a ≈ ret v , then run a
∗

→ v .

The monadic translation Correctness

Correctness of the monadic translation

Theorem 3

If a⇒ v, then [[a]] ≈ ret [[v ]]v .

The proof is by induction on a derivation of a⇒ v and case analysis on
the last evaluation rule.

The cases a = N, a = x and a = λx .b are obvious: we have a = v ,
therefore [[a]] = ret [[v ]]v .

The monadic translation Correctness

Correctness of the monadic translation

For the let case:
b ⇒ v ′ c[x ← v ′]⇒ v

let x = b in c ⇒ v

The following equivalences hold:

[[a]] = bind [[b]] (λx .[[c]])

(ind.hyp + prop.3) ≈ bind (ret [[v ′]]v ) (λx .[[c]])

(prop.2) ≈ [[c]][x ← [[v ′]]v = [[c[x ← v ′]]]

(ind.hyp.) ≈ ret [[v ]]v

The monadic translation Correctness

Correctness of the monadic translation

For the application case:

b ⇒ λx .d c ⇒ v ′ d [x ← v ′]⇒ v

b c ⇒ v

The following equivalences hold:

[[a]] = bind [[b]] (λy .bind [[c]] (λz . y z))

(ind.hyp + prop.3) ≈ bind (ret (λx .[[d ]])) (λy .bind [[c]] (λz . y z))

(prop.2) ≈ bind [[c]] (λz . (λx .[[d ]]) z))

(ind.hyp + prop.3) ≈ bind (ret [[v ′]]v (λz . (λx .[[d ]]) z))

(prop.2) ≈ (λx .[[d ]]) [[v ′]]v

(prop.1) ≈ [[d ]][x ← [[v ′]]v ] = [[d [x ← v ]]]

(ind.hyp.) ≈ ret [[v ]]v

The monadic translation Correctness

Correctness of the monadic translation

Theorem 4

If a⇒ N, then run [[a]]
∗

→ N.

Proof.

Follows from theorem 3 and property 4 of ≈.

Note that we proved this theorem only for pure terms a that do not use
monad-specific constructs. These constructs add more cases, but often the
proof cases for application, etc, are unchanged. (Exercise.)

The monadic translation Application to some monads

Application to the Exception monad

Define a1 ≈ a2 as ∃a, a1
∗

→ a
∗

← a2.

Some interesting properties of this relation:

If a→ a′ then a ≈ a′.

If a ≈ a′ and a
∗

→ v , then a′
∗

→ v .

It is transitive, for if a1
∗

→ a
∗

← a2
∗

→ a′
∗

← a3, determinism of the →
reduction implies that either a

∗

→ a′ or a′
∗

→ a. In the former case,
a1

∗

→ a′
∗

← a3, and in the latter case, a1
∗

→ a
∗

← a3.

It is compatible with reduction contexts: E [a1] ≈ E [a2] if a1 ≈ a2 and
E is a reduction context.

We now check that ≈ satisfies the hypothesis of theorem 3.

The monadic translation Application to some monads

Application to the Exception monad

1 (λx .a) v ≈ a[x ← v ]
Trivial since (λx .a) v → a[x ← v ].

2 bind (ret v) (λx .b) ≈ b[x ← v ]. We have

bind (ret v) (λx .b)

→ bind (V (v)) (λx .b)
∗

→ match V (v) with E (y)→ y | V (z)→ (λx .b) z

→ (λx .b) v → b[x ← v ]

3 bind a1 (λx .b) ≈ bind a2 (λx .b) if a1 ≈ a2.
Trivial since bind [ ] (λx .b) is an evaluation context.

4 If a ≈ ret v , then run a
∗

→ v .
Since ret v

∗

→ V (v), we have a
∗

→ V (v) and the result follows.

The monadic translation Application to some monads

Application to the Continuation monad

Define a1 ≈ a2 as ∀k ∈ Values, ∃a, a1 k
∗

→ a
∗

← a2 k .

1 (λx .a) v ≈ a[x ← v ]
Trivial since (λx .a) v k → a[x ← v ] k .

2 bind (ret v) (λx .b) ≈ b[x ← v ]. We have

bind (ret v) (λx .b) k → bind (λk ′. k ′ v) (λx .b)
∗

→ (λk ′. k ′ v) (λy . (λx .b) y k)

→ (λy . (λx .b) y k) v

→ (λx .b) v k)

→ b[x ← v ] k

The monadic translation Application to some monads

Application to the Continuation monad

1 bind a1 (λx .b) ≈ bind a2 (λx .b) if a1 ≈ a2

We have bind ai (λx .b) k
∗

→ ai (λv . (λx .b) v k) for i = 1, 2.
Using the hypothesis a1 ≈ a2 with the continuation (λv . (λx .b) v k),

we obtain a term a such that ai (λv . (λx .b) v k)
∗

→ a for i = 1, 2.

Therefore, bind ai (λx .b) k
∗

→ a for i = 1, 2, and the result follows.

2 If a ≈ ret v , then run a
∗

→ v .
The result follows from ret v (λx .x)

∗

→ v .

The monadic translation Application to some monads

Application to the State monad

Define a1 ≈ a2 as ∀s ∈ Values, ∃a, a1 s
∗

→ a
∗

← a2 s.

The proofs of hypotheses 1–4 are similar to those for exceptions.

Monadic programming

Monadic programming

Monads as a general programming technique

Monads provide a systematic way to structure programs into two
well-separated parts:

the algorithms proper, and

the “plumbing” of computations needed by these algorithms (state
passing, exception handling, non-deterministic choice, etc).

In addition, monads can also be used to modularize code and offer new
possibilities for reuse:

Code in monadic form can be parameterized over a monad and reused
with several monads.

Monads themselves can be built in an incremental manner.

Monadic programming More examples of monads

The Logging monad (a.k.a. the Writer monad)

Enables computations to log messages. A special case of the State monad,
guaranteeing that the log grows monotonically.

module Log = struct

type log = string list

type α mon = log → α× log

let ret a = fun l -> (a, l)

let bind m f = fun l -> match m l with (x, l’) -> f x l’

let run m = match m [] with (x, l) -> (x, List.rev l)

let log msg = fun l -> ((), msg :: l)

end

Monadic programming More examples of monads

Example of use

Before monadic translation:

let abs n =

if n >= 0

then (log "positive"; n)

else (log "negative"; -n)

After monadic translation:

let abs n =

if n >= 0

then Log.bind (Log.log "positive") (fun _ -> Log.ret n)

else Log.bind (Log.log "negative") (fun _ -> Log.ret (-n))

Monadic programming More examples of monads

Non-determinism, a.k.a. the List monad

Provides computations with non-deterministic choice as well as failure.
Underneath, computes the list of all possible results.

module Nondet = struct

type α mon = α list

let ret a = a :: []

let rec bind m f =

match m with [] -> [] | hd :: tl -> f hd @ bind tl f

let run m = match m with hd :: tl -> hd

let runall m = m

let fail = []

let either a b = a @ b

end

Monadic programming More examples of monads

Example of use
All possible ways to insert an element x in a list l:

let rec insert x l =

Nondet.either (Nondet.ret (x :: l))

(match l with

| [] -> Nondet.fail

| hd :: tl ->

Nondet.bind (insert x tl)

(fun l’ -> Nondet.ret (hd :: l’)))

All permutations of a list l:

let rec permut l =

match l with

| [] -> Nondet.ret []

| hd :: tl ->

Nondet.bind (permut tl) (fun l’ -> insert hd l’)

Monadic programming More examples of monads

Monads for randomized computations

Consider a source language with randomized constructs such as

rand n return a uniformly-distributed integer in [0, n[

choose p a b evaluate a with probability p ∈ [0, 1]
evaluate b with probability 1− p

In a monadic interpretation, these constructs have type

rand : int→ int mon

choose : ∀α. float→ α mon→ α mon→ α mon

Monadic programming More examples of monads

Examples of randomized computations

let roll_3d6 =

M.bind (M.rand 6) (fun d1 ->

M.bind (M.rand 6) (fun d2 ->

M.bind (M.rand 6) (fun d3 ->

M.return (1+d1 + 1+d2 + 1+d3))))

let traffic_light =

M.choose 0.05 (M.return Yellow)

(M.choose 0.5 (M.return Red)

(M.return Green))

Monadic programming More examples of monads

First implementation: the Simulation monad

Uses a pseudo-random number generator to give values to random
variables (Monte-Carlo simulation). This is a variant of the State monad.

module Random_Simulation = struct

type α mon = int → α× int

let ret a = fun s -> (a, s)

let bind m f = fun s -> match m s with (x, s) -> f x s

let next_state s = s * 25173 + 1725

let rand n = fun s -> ((abs s) mod n, next_state s)

let choose p a b = fun s ->

if float (abs s) <= p *. float max_int

then a (next_state s) else b (next_state s)

end

Monadic programming More examples of monads

Second implementation: the Distribution monad
With the same interface, this monad computes the distribution of the
results: all possible result values along with their probabilities.
This is an extension of the List monad.

module Random_Distribution = struct

type α mon = (α× float) list

let ret a = [(a, 1.0)]

let bind m f =

[ (y, p1 *. p2) | (x, p1) <- m, (y, p2) <- f x ]

let rand n = [ (0, 1
n
); ...; (n-1, 1

n
) ]

let choose p a b =

[ (x, p *. p1) | (x, p1) <- a ] @

[ (x, (1.0 -. p) *. p2) | (x, p2) <- b ]

end

Monadic programming More examples of monads

Third implementation: the Expectation monad

Still with the same interface, this monad computes the expectation of a
result (of type α) w.r.t. a given measure (a function α→ float).
This is an extension of the Continuation monad.

module Random_Expectation = struct

type α mon = (α -> float) -> float

let ret x = fun k -> k x

let bind x f = fun k -> x (fun vx -> f vx k)

let rand n = fun k -> 1
n
*. k 0 +. ... +. 1

n
*. k (n-1)

let choose p a b = fun k -> p *. a k +. (1.0 -. p) *. b k

end

Monadic programming Monad transformers

Combining monads
What if we need both exceptions and state in an algorithm?

We can write (from scratch) a monad that supports both. Notice that
there are several choices:

type α mon = state→ (α× state) outcome

I.e. the state is discarded when we raise an exception.

type α mon = state→ α outcome× state

I.e. the state is kept when we raise an exception.

In the second case, trywith can be defined in two ways:

trywith m f = λs. match m s with

| (V (v), s ′)→ (V (v), s ′)

| (E (e), s ′)→ f e

(

s

s ′

)

The s choice backtracks the assignments made by the computation m;
the s ′ choice preserves them.

Monadic programming Monad transformers

Monad transformers

A more systematic way to build combined monads is to use monad
transformers.

A monad transformer takes any monad M and returns a monad M ′ with
additional capabilities, e.g. exceptions, state, continuation. It also provides
a lift function that transforms M computations (of type α M.mon) into
M ′ computations (of type α M ′.mon)

In Caml, monad transformers are naturally presented as functors, i.e.
functions from modules to modules. (Haskell uses type classes.)

Monadic programming Monad transformers

Signature for monads

The Caml module signature for a monad is:

module type MONAD = sig

type α mon

val ret: α -> α mon

val bind: α mon -> (α -> β mon) -> β mon

val run: α mon -> α

end

X. Leroy (INRIA) Functional programming languages MPRI 2-4-2, 2007 43 / 52

Monadic programming Monad transformers

The Identity monad

The Identity monad is a trivial instance of this signature:

module Identity = struct

type α mon = α

let ret x = x

let bind m f = f m

let run m = m

end

Monadic programming Monad transformers

Monad transformer for exceptions

module ExceptionTransf(M: MONAD) = struct

type α outcome = V of α | E of exn

type α mon = (α outcome) M.mon

let ret x = M.ret (V x)

let bind m f =

M.bind m (function E e -> M.ret (E e) | V v -> f v)

let lift x = M.bind x (fun v -> M.ret (V v))

let run m = M.run (M.bind m (function V x -> M.ret x))

let raise e = M.ret (E e)

let trywith m f =

M.bind m (function E e -> f e | V v -> M.ret (V v))

end

Monadic programming Monad transformers

Monad transformer for state

module StateTransf(M: MONAD) = struct

type α mon = state -> (α * state) M.mon

let ret x = fun s -> M.ret (x, s)

let bind m f =

fun s -> M.bind (m s) (fun (x, s’) -> f x s’)

let lift m = fun s -> M.bind m (fun x -> M.ret (x, s))

let run m =

M.run (M.bind (m empty_store) (fun (x, s’) -> M.ret x))

let ref x = fun s -> M.ret (store_alloc x s)

let deref r = fun s -> M.ret (store_read r s, s)

let assign r x = fun s -> M.ret (store_write r x s)

end

Monadic programming Monad transformers

Monad transformer for continuations

module ContTransf(M: MONAD) = struct

type α mon = (α -> answer M.mon) -> answer M.mon

let ret x = fun k -> k x

let bind m f = fun k -> m (fun v -> f v k)

let lift m = fun k -> M.bind m k

let run m = M.run (m (fun x -> M.ret x))

let callcc f = fun k -> f k k

let throw c x = fun k -> c x

end

Using monad transformers

module StateAndException = struct

include ExceptionTransf(State)

let ref x = lift (State.ref x)

let deref r = lift (State.deref r)

let assign r x = lift (State.assign r x)

end

This gives a type α mon = state→ α outcome× state,
i.e. state is preserved when raising exceptions.

The other combination, StateTransf(Exception) gives
α mon = state→ (α× state) outcome,
i.e. state is discarded when an exception is raised.

Monadic programming Monad transformers

The Concurrency monad transformer

Generalizing the Continuation monad transformer, we can define
concurrency (interleaving of atomic computations) as follows:

module Concur(M: MONAD) = struct

type answer =

| Seq of answer M.mon

| Par of answer * answer

| Stop

type α mon = (α -> answer) -> answer

let ret x = fun k -> k x

let bind x f = fun k -> x (fun v -> f v k)

let atom m = fun k -> Seq(M.bind m (fun v -> M.ret (k v)))

let stop = fun k -> Stop

let par m1 m2 = fun k -> Par (m1 k, m2 k)

Monadic programming Monad transformers

The Concurrency monad transformer

If m : α mon, applying m to the initial continuation λx , Stop builds a tree
of computations such as:

Seq m1

Par

Seq m2

Seq m4

Stop

Seq m3

Par

Seq m5

Stop

Seq m6

Stop

All that remains is to execute the atomic actions m1, . . . ,m6 in
breadth-first order, simulating interleaved execution.

Monadic programming Monad transformers

The Concurrency monad transformer

module Concur(M: MONAD) = struct

...

let rec schedule acts =

match acts with

| [] -> M.ret ()

| Seq m :: rem ->

M.bind m (fun m’ -> schedule (rem @ [m’]))

| Par(a1, a2) :: rem ->

schedule (a1 :: a2 :: rem)

| Stop :: rem ->

schedule rem

let run m = M.run (schedule [m (fun _ -> Stop)])

end

Monadic programming Monad transformers

Example of use

module M = Concur(Log)

let rec loop n s =

if n <= 0

then M.ret ()

else M.bind (M.atom (Log.log s)) (fun _ -> loop (n-1) s)

M.run (M.bind (M.atom (Log.log "start:")) (fun _ ->

M.par (loop 6 "a") (loop 4 "b")))

This code will log “start:ababababaaaa”

